NAG Library Routine Document

G03BDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G03BDF calculates a ProMax rotation, given information following an orthogonal rotation.

2 Specification

```
 SUBROUTINE G03BDF(STAND, N, M, X, LDX, RO, LDRO, POWER, FP, LDFP, R,
1                  LDR, PHI, LDPHI, FS, LDFS, IFAIL)
 INTEGER           N, M, LDX, LDRO, LDFP, LDR, LDPHI, LDFS, IFAIL
 double precision  X(LDX,M), RO(LDRO,M), POWER, FP(LDFP,M), R(LDR,M),
1                  PHI(LDPHI,M), FS(LDFS,M)
 CHARACTER*1       STAND
```

3 Description

Let X and Y denote n by m matrices each representing a set of n points in an m-dimensional space. The X matrix is a matrix of loadings as returned by G03BAF, that is following an orthogonal rotation of a loadings matrix Z. The target matrix Y is calculated as a power transformation of X that preserves the sign of the loadings. Let X_{ij} and Y_{ij} denote the (i,j)th element of matrices X and Y. Given a value greater than one for the exponent p:

$$Y_{ij} = \delta_{ij} \left\| X_{ij} \right\|^p,$$

for

$$i = 1, 2, \dots, n;$$

$$j = 1, 2, \dots, m;$$

$$\delta_{ij} = \begin{cases} -1, \text{ if } X_{ij} < 0; \\ 1, \text{ otherwise.} \end{cases}$$

The above power transformation tends to increase the difference between high and low values of loadings and is intended to increase the interpretability of a solution.

In the second step a solution of:

$$XW = Y, \quad X, Y \in \mathbb{R}^{\mathrm{n} \times \mathrm{m}}, W \in \mathbb{R}^{\mathrm{m} \times \mathrm{m}},$$

is found for W in the least squares sense by use of singular value decomposition of the orthogonal loadings X. The ProMax rotation matrix R is then given by

$$R = OW\tilde{W}, \quad O, \tilde{W} \in \mathbb{R}^{\mathrm{m} \times \mathrm{m}},$$

where O is the rotation matrix from an orthogonal transformation, and $\tilde{W}$ is a matrix with the square root of diagonal elements of $\left(W^{\mathrm{T}}W\right)^{-1}$ on its diagonal and zeros elsewhere.

The ProMax factor pattern matrix P is given by

$$P = XW\tilde{W}, \quad P \in \mathbb{R}^{\mathrm{n} \times \mathrm{m}};$$

the inter-factor correlations Φ are given by

$$\Phi = \left(Q^{\mathrm{T}}Q\right)^{-1}, \quad \Phi \in \mathbb{R}^{\mathrm{m} \times \mathrm{m}};$$

where $Q = W\tilde{W}$; and the factor structure S is given by

$$S = P\Phi, \quad S \in \mathbb{R}^{n \times m}.$$

Optionally, the rows of target matrix Y can be scaled by the communalities of loadings.

4 References

None.

5 Parameters

1: STAND – CHARACTER*1 *Input*

On entry: indicates how loadings are normalized.

STAND = 'S'
: Rows of Y are (Kaiser) normalized by the communalities of the loadings.

STAND = 'U'
: Rows are not normalized.

Constraint: STAND = 'U' or 'S'.

2: N – INTEGER *Input*

On entry: n the number of points.

Constraint: N $\geq$ M.

3: M – INTEGER *Input*

On entry: m, the number of dimensions.

Constraint: M $\geq$ 1.

4: X(LDX,M) – ***double precision*** array *Input*

On entry: the loadings matrix following an orthogonal rotation, X.

5: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G03BDF is called.

Constraint: LDX $\geq$ N.

6: RO(LDRO,M) – ***double precision*** array *Input*

On entry: the orthogonal rotation matrix, O.

7: LDRO – INTEGER *Input*

On entry: the first dimension of the array RO as declared in the (sub)program from which G03BDF is called.

Constraint: LDRO $\geq$ M.

8: POWER – ***double precision*** *Input*

On entry: p, the value of exponent.

Constraint: POWER $>$ 1.0.

9: FP(LDFP,M) – ***double precision*** array *Output*

On exit: the factor pattern matrix, P.

10: LDFP – INTEGER *Input*

On entry: the first dimension of the array FP as declared in the (sub)program from which G03BDF is called.

Constraint: LDFP $\geq$ N.

11: R(LDR,M) – ***double precision*** array *Output*

On exit: the ProMax rotation matrix, R.

12: LDR – INTEGER *Input*

On entry: the first dimension of the array R as declared in the (sub)program from which G03BDF is called.

Constraint: LDR $\geq$ M.

13: PHI(LDPHI,M) – ***double precision*** array *Output*

On exit: the matrix of inter-factor correlations, Φ.

14: LDPHI – INTEGER *Input*

On entry: the first dimension of the array PHI as declared in the (sub)program from which G03BDF is called.

Constraint: LDPHI $\geq$ M.

15: FS(LDFS,M) – ***double precision*** array *Output*

On exit: the factor structure matrix, S.

16: LDFS – INTEGER *Input*

On entry: the first dimension of the array FS as declared in the (sub)program from which G03BDF is called.

Constraint: LDFS $\geq$ N.

17: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, STAND $\neq$ 'U' or 'S',
or M < 1,
or POWER ≤ 1.0.

IFAIL = 2

On entry, N < M,
or LDX < N,
or LDRO < M,
or LDFP < N,
or LDFS < N,
or LDR < M,
or LDPHI < M.

IFAIL = 20

An SVD decomposition failed to converge.

IFAIL = 100

An internal error occurred in the calculations. Check array declarations and calls to G03BDF. Please contact NAG.

7 Accuracy

The calculations are believed to be stable.

8 Further Comments

None.

9 Example

This example reads a loadings matrix and calculates a varimax transformation before calculating P, R and σ for a ProMax rotation.

9.1 Program Text

```
*     G03BDF Example Program Text
*     Mark 22 Release. NAG Copyright 2008.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, MMAX
      PARAMETER        (NMAX=44,MMAX=11)
*     .. Local Scalars ..
      DOUBLE PRECISION POWER
      INTEGER          I, IFAIL, J, LDFP, LDFS, LDPHI, LDR, LDRO, LDX,
     +                 M, N
      CHARACTER        STAND
*     .. Local Arrays ..
      DOUBLE PRECISION FP(NMAX,MMAX), FS(NMAX,MMAX), PHI(MMAX,MMAX),
     +                 R(MMAX,MMAX), RO(MMAX,MMAX), X(NMAX,MMAX)
*     .. External Subroutines ..
      EXTERNAL         G03BDF, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G03BDF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) STAND, N, M, POWER
      IF (N.GT.NMAX .OR. M.GT.MMAX) THEN
         WRITE (NOUT,99998)
         GO TO 60
      END IF
      LDX = NMAX
      LDRO = MMAX
      LDFP = NMAX
      LDFS = NMAX
      LDR = MMAX
```

```
      LDPHI = MMAX
*     Read orthognal loadings matrix.
      DO 20 I = 1, N
         READ (NIN,*) (X(I,J),J=1,M)
   20 CONTINUE
*     Read orthogonal rotation matrix.
      DO 40 I = 1, M
         READ (NIN,*) (RO(I,J),J=1,M)
   40 CONTINUE
*
      IFAIL = 1
      CALL G03BDF(STAND,N,M,X,LDX,RO,LDRO,POWER,FP,LDFP,R,LDR,PHI,LDPHI,
     +            FS,LDFS,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99999) IFAIL
         GO TO 60
      END IF
*
      WRITE (NOUT,*)
      CALL X04CAF('G','X',N,M,FP,LDFP,'Factor pattern',IFAIL)
      WRITE (NOUT,*)
      CALL X04CAF('G','X',M,M,R,LDR,'ProMax rotation',IFAIL)
      WRITE (NOUT,*)
      CALL X04CAF('G','X',M,M,PHI,LDPHI,'Inter-factor correlations',
     +            IFAIL)
      WRITE (NOUT,*)
      CALL X04CAF('G','X',N,M,FS,LDFS,'Factor structure',IFAIL)
*
   60 CONTINUE
*
99999 FORMAT (1X,/1X,' ** G03BDF returned with IFAIL = ',I5)
99998 FORMAT (1X,' ** Problem size too large, increase array limits')
      END
```

9.2 Program Data

```
G03BDF Example Program Data
's' 5 2 3.0 : STAND N M POWER
 0.94072 -0.00004
 0.90419  0.00055
 0.79085  0.41509
 0.02255  0.98874
 0.14625  0.97499 : End of factor pattern matrix from orthogonal rotation
 0.78895  0.61446
-0.61446  0.78895 : End of orthogonal transform (rotation) matrix
```

9.3 Program Results

```
 G03BDF Example Program Results

 Factor pattern
             1          2
 1      0.9556    -0.0979
 2      0.9184    -0.0935
 3      0.7605     0.3393
 4     -0.0791     1.0019
 5      0.0480     0.9751

 ProMax rotation
          1       2
 1   0.7380  0.5420
 2  -0.7055  0.8653

 Inter-factor correlations
          1       2
 1   1.0000  0.2019
 2   0.2019  1.0000

 Factor structure
          1       2
```

```
 1   0.9358  0.0950
 2   0.8995  0.0919
 3   0.8290  0.4929
 4   0.1232  0.9860
 5   0.2448  0.9848
```